
AIMMS Language Reference - Parameter Declaration

This file contains only one chapter of the book. For a free download of the complete book in pdf format, please visit www.aimms.com or order your hard-copy at www.lulu.com/aimms.

Copyright © 1993–2011 by Paragon Decision Technology B.V. All rights reserved.

Paragon Decision Technology B.V.	Paragon Decision Technology Inc.	Paragon Decision Technology Pte.
Schipholweg 1	500 108th Avenue NE	Ltd.
2034 LS Haarlem	Ste. # 1085	80 Raffles Place
The Netherlands	Bellevue, WA 98004	UOB Plaza 1, Level 36-01
Tel.: +31 23 5511512	USA	Singapore 048624
Fax: +31 23 5511517	Tel.: +1 425 458 4024	Tel.: +65 9640 4182
	Fax: +1 425 458 4025	

Email: info@aimms.com
WWW: www.aimms.com

AIMMS is a registered trademark of Paragon Decision Technology B.V. IBM ILOG CPLEX and sc CPLEX is a registered trademark of IBM Corporation. GUROBI is a registered trademark of Gurobi Optimization, Inc. KNITRO is a registered trademark of Ziena Optimization, Inc. XPRESS-MP is a registered trademark of FICO Fair Isaac Corporation. MOSEK is a registered trademark of Mosek ApS. WINDOWS and EXCEL are registered trademarks of Microsoft Corporation. $\text{T}_{\text{E}}\text{X}$, $\text{L}_{\text{A}}\text{T}_{\text{E}}\text{X}$, and $\text{A}_{\text{M}}\text{S}_{\text{L}}\text{A}_{\text{T}}\text{E}_{\text{X}}$ are trademarks of the American Mathematical Society. LUCIDA is a registered trademark of Bigelow & Holmes Inc. ACROBAT is a registered trademark of Adobe Systems Inc. Other brands and their products are trademarks of their respective holders.

Information in this document is subject to change without notice and does not represent a commitment on the part of Paragon Decision Technology B.V. The software described in this document is furnished under a license agreement and may only be used and copied in accordance with the terms of the agreement. The documentation may not, in whole or in part, be copied, photocopied, reproduced, translated, or reduced to any electronic medium or machine-readable form without prior consent, in writing, from Paragon Decision Technology B.V.

Paragon Decision Technology B.V. makes no representation or warranty with respect to the adequacy of this documentation or the programs which it describes for any particular purpose or with respect to its adequacy to produce any particular result. In no event shall Paragon Decision Technology B.V., its employees, its contractors or the authors of this documentation be liable for special, direct, indirect or consequential damages, losses, costs, charges, claims, demands, or claims for lost profits, fees or expenses of any nature or kind.

In addition to the foregoing, users should recognize that all complex software systems and their documentation contain errors and omissions. The authors, Paragon Decision Technology B.V. and its employees, and its contractors shall not be responsible under any circumstances for providing information or corrections to errors and omissions discovered at any time in this book or the software it describes, whether or not they are aware of the errors or omissions. The authors, Paragon Decision Technology B.V. and its employees, and its contractors do not recommend the use of the software described in this book for applications in which errors or omissions could threaten life, injury or significant loss.

This documentation was typeset by Paragon Decision Technology B.V. using $\text{L}_{\text{A}}\text{T}_{\text{E}}\text{X}$ and the LUCIDA font family.

Chapter 4

Parameter Declaration

The word parameter does not have a uniform meaning in the scientific community. When you are a statistician, you are likely to view a parameter as an unknown quantity to be estimated from observed data. *In AIMMS the word parameter denotes a known quantity that holds either numeric or string-valued data.* In programming languages the term variable is used for this purpose. However, this is not the convention adopted in AIMMS, where, in the context of a mathematical program, the word variable is reserved for an unknown quantity. Outside this context, a variable behaves as if it were a parameter. The terminology in AIMMS is consistent with the standard operations research terminology that distinguishes between parameters and variables.

Terminology

Rather than putting the explicit data values directly into your expressions, it is a much better practice to group these values together in parameters and to write all your expressions using these symbolic parameters. Maintaining a model that contains explicit data is a painstaking task and error prone, because the meaning of each separate number is not clear. Maintaining a model in symbolic form, however, is much easier and frequently boils down to simply adjusting the data of a few clearly named parameters at a single point.

Why use parameters

Consider the set `Cities` introduced in the previous chapter and a parameter `FixedTransport(i,j)`. Suppose that the cost of each unit of transport between cities `i` and `j` is stored in the parameter `UnitTransportCost(i,j)`. Then the definition of `TotalTransportCost` can be expressed as

Example

```
TotalTransportCost := sum[(i,j), UnitTransportCost(i,j)*FixedTransport(i,j)];
```

Not only is this expression easy to understand, it also makes your model extendible. For instance, an extra city can be added to your model by simply adding an extra element to the set `Cities` as well as updating the tables containing the data for the parameters `UnitTransportCost` and `FixedTransport`. After these changes the above statement will automatically compute `TotalTransportCost` based on the new settings without any explicit change to the symbolic model formulation.

4.1 PARAMETER declaration and attributes

Parameters in AIMMS can hold data of the following four data types:

- numeric values,
- strings,
- set elements, and
- unit expressions.

Declaration and attributes

Prior to declaring a parameter in the model editor you need to decide on its data type. In the model tree parameters of each type have their own icon. The attributes of parameters are given in Table 4.1.

Attribute	Value-type	See also page
INDEX DOMAIN	<i>index-domain</i>	
RANGE	<i>range</i>	
DEFAULT	<i>constant-expression</i>	
UNIT	<i>unit-expression</i>	
PROPERTY	NoSave, Stochastic, Uncertain, Random, <i>numeric-storage-property</i>	45
TEXT	<i>string</i>	19
COMMENT	<i>comment string</i>	19, 31
DEFINITION	<i>expression</i>	33
UNCERTAINTY	<i>expression</i>	46, 291
REGION	<i>expression</i>	46, 288
DISTRIBUTION	<i>expression</i>	46, 294

Table 4.1: PARAMETER attributes

The following declarations demonstrate some basic numeric parameter declarations

Basic examples

```
PARAMETER:
  identifier : Population
  index domain : i
  range      : [0,inf)
  unit       : [ 1000 ]
  text       : Population of city i in thousands ;

PARAMETER:
  identifier : Distance
  index domain : (i,j)
  range      : [0,inf)
  unit       : [ km ]
  text       : Distance from city i to city j in km ;
```

For each multidimensional identifier you need to specify its dimensions by providing a list of index bindings at the INDEX DOMAIN attribute. Identifiers without an INDEX DOMAIN are said to be *scalar*. In the index domain you can specify default or local bindings to either simple or compound sets. The totality of dimensions of all bindings determine the total dimension of the identifier. Any references outside the index domain, either through execution statements or from within the graphical user interface are skipped.

*The INDEX
DOMAIN attribute*

You can also use the INDEX DOMAIN attribute to specify a logical expression which further restricts the valid tuples in the domain. During execution, assignments to tuples that do not satisfy the domain condition are ignored. Also, evaluation of references to such tuples in expressions will result in the value zero. Note that, if the domain condition contains references to other data in your model, the set of valid tuples in the domain may change during a single interactive session.

*Domain
condition*

Consider the sets `ConnectedCities` with default index `cc` and `DestinationCitiesFromSupply(i)` from the previous chapter. The following statements illustrate a number of possible declarations of the two-dimensional identifier `UnitTransportCost` with varying index domains.

Example

```
PARAMETER:
  identifier   : UnitTransportCost
  index domain : (i,j) ;
PARAMETER:
  identifier   : UnitTransportCostWithCondition
  index domain : (i,j) in ConnectedCities ;
PARAMETER:
  identifier   : UnitTransportCostWithCompoundDomain
  index domain : cc ;
PARAMETER:
  identifier   : UnitTransportCostWithIndexedDomain
  index domain : (i, j in DestinationCitiesFromSupply(i)) ;
```

The identifiers defined in the previous example will behave as follows.

Explanation

- The identifier `UnitTransportCost` is defined over the full Cartesian product $Cities \times Cities$ by means of the default bindings of the indices `i` and `j`. You will be able to assign values to every pair of cities (i,j) , even though there is no connection between them.
- The identifier `UnitTransportCostWithCondition` is defined over the same Cartesian product of sets. Its domain, however, is restricted by an additional condition $(i,j) \text{ in } ConnectedCities$ which will exclude assignments to tuples that do not satisfy this condition, or evaluate to zero when referenced.
- The identifier `UnitTransportCostWithCompoundDomain` is defined over the two-dimensional compound set `ConnectedCities` by the default binding of the index `cc`. Although the declaration seems equivalent to that of

UnitTransportCostWithCondition, AIMMS will now produce a domain error, when you try to make assignments to tuples outside of this set.

- Finally, the identifier UnitTransportCostWithIndexedDomain is defined over a subset of the Cartesian product $Cities \times Cities$. The second element j must lie in the subset $DestinationCities(i)$ associated with i . AIMMS will produce a domain error if this condition is not satisfied.

For every indexed identifier you have access to the current contents of the INDEX DOMAIN attribute through the .Domain suffix associated with the identifier at hand. The .Domain suffix behaves like a set of the dimension of its associated identifier. You can use it, for instance, in the INDEX DOMAIN of another identifier to indicate that two identifiers should have exactly the same domain.

The .Domain suffix

The following declaration illustrates a parameter that has exactly the same domain as the parameter UnitTransportCostWithIndexedDomain declared above.

Example

```
PARAMETER:
  identifier   : UnitTransportCostWithSharedDomain
  index domain : (i,j) in UnitTransportCostWithIndexedDomain.Domain ;
```

With the RANGE attribute you can restrict the values to certain intervals or sets. The possible values for the RANGE attribute are:

The RANGE attribute

- one of the predefined ranges Real, Nonnegative, Nonpositive, Integer, or Binary,
- any one of the interval expressions $[a, b]$, $[a, b)$, $(a, b]$, or (a, b) , where a square bracket implies inclusion into the interval and a round bracket implies exclusion,
- any enumerated integer set expression, e.g. $\{a .. b\}$ covering all integers from a until and including b ,
- a set identifier, if you want the values to be elements of that set. For set element-valued parameters this entry is mandatory.

The values for a and b can be a constant number, inf, -inf, or a parameter reference involving some or all of the indices on the index domain of the declared identifier.

Consider the following declarations.

Example

```
PARAMETER:
  identifier   : UnitTransportCost
  index domain : (i,j)
  range       : [ UnitLoadingCost(i), 100 ] ;
PARAMETER:
  identifier   : DefaultUnitsShipped
  index domain : (i,j)
  range       : { MinShipment(i) .. MaxShipment(j) } ;
```

It limits the values of the identifier `UnitTransportCost(i, j)` to an interval from `UnitLoadingCost(i)` to 100. Note that the lower bound of the interval has a smaller dimension than the identifier itself. The integer identifier `Default-UnitsShipped(i, j)` is limited to an integer range through an enumerated integer range inside the set brackets.

In AIMMS, parameters that have not been assigned an explicit value are given a default value automatically. You can specify the default value with the `DEFAULT` attribute. The value of this attribute *must* be a constant expression. If you do not provide a default value for the parameter, AIMMS will assume the following defaults:

The DEFAULT attribute

- 0 for numbers,
- 1 for unit-valued parameters,
- the empty string "" for strings, and
- the empty element '' for set elements.

The definition attribute of a parameter can contain a valid (indexed) numerical expression. Whenever a defined parameter is referenced inside your model, AIMMS will, by default, recompute the associated data if (data) changes to any of the identifiers referenced in its definition make its current data out-of-date. In the definition expression you can refer to any of the indices in the index domain as if the definition was the right-hand side of an assignment statement to the parameter at hand (see also Section 8.2).

The DEFINITION attribute

The following declaration illustrates an indexed `DEFINITION` attribute.

Example

```
PARAMETER:
  identifier   : MaxTransportFrom
  index domain : i
  definition   : Max(j, Transport(i,j));
```

Whenever you provide a definition for an *indexed* parameter, you should carefully verify whether and how that parameter is used in the context of one of AIMMS' loop statements (see also Section 8.3). When, due to changes in only a slice of the dependent data of a definition during a previous iteration, AIMMS (in fact) only needs to evaluate a single slice of a defined parameter during the actual iteration, you should probably not be using a defined parameter. AIMMS' automatic evaluation scheme for defined identifiers will always recompute the data for such identifiers *for the whole domain of definition*, which can lead to severe inefficiencies for high-dimensional defined parameters. You can find a more detailed discussion on this issue in Section 13.2.3.

Care when used in loops

By associating a UNIT to every numerical identifier in your model, you can let AIMMS help you check your model's consistency. AIMMS also uses the UNIT attribute when presenting data and results in both the output files of a model and the graphical user interface. You can find more information on the use of units in Chapter 30.

The UNIT attribute

The PROPERTY attribute can hold various properties of the identifier at hand. The allowed properties for a parameter are NoSave or one of the numerical storage properties Integer, Integer32, Integer16, Integer8 or Double, in addition to the properties Stochastic, Uncertain, Random which are discussed in Section 4.1.1.

The PROPERTY attribute

- The property NoSave indicates whether the identifier values are stored in cases. It is discussed in detail in Section 3.2.
- By default, the values of numeric parameters are stored as double precision floating point numbers. By specifying one of the storage properties Integer, Integer32, Integer16, Integer8, or Double AIMMS will store the values of the identifier as (signed) integers of default machine length, 4 bytes, 2 bytes or 1 byte, or as a double precision floating point number respectively. These properties are only applicable to parameters with an integer range.

During execution you can change the properties of a parameter through the PROPERTY statement. The syntax of the PROPERTY statement and examples of its use can be found in Section 8.5.

The PROPERTY statement

With the TEXT attribute you can provide one line of descriptive text for the end-user. If the TEXT string of an indexed parameter or variable contains a reference to one or more indices in the index domain, then the corresponding elements are substituted for these indices in any display of the identifier text.

The TEXT attribute

4.1.1 Properties and attributes for uncertain data

The AIMMS modeling language allows you to specify both stochastic programs and robust optimization models. Both methodologies are designed to deal with models involving data uncertainty. In stochastic programming the uncertainty is expressed by specifying multiple scenarios, each of which can define scenario-specific values for certain parameters in your model. Stochastic programming is discussed in full detail in Chapter 19. For robust optimization, parameters can be declared to not have a single fixed value, but to take their values from an user-defined uncertainty set. Robust optimization is discussed in Chapter 20.

Stochastic programming and robust optimization

The following PARAMETER properties are available in support of stochastic programming and robust optimization models.

Properties

- The property `Stochastic` indicates that the identifier can hold stochastic event data for a stochastic model. It is discussed in detail in Section 19.2.
- The property `Uncertain` indicates that the identifier can hold uncertain values from an uncertainty set specified through the `UNCERTAINTY` and/or `REGION` attributes. Uncertain parameters are used in AIMMS' robust optimization facilities, and are discussed in detail in Section 20.2.
- The property `Random` indicates that the identifier can hold random values with respect to a distribution with characteristics specified through the `DISTRIBUTION` attribute. Random parameters are used in AIMMS' robust optimization facilities, and are discussed in detail in Section 20.3.

The `UNCERTAINTY` and `REGION` attributes are available if the parameter at hand has been declared uncertain using the `Uncertain` property. Uncertain parameters are used by AIMMS' robust optimization framework, and are discussed in full detail in Section 20.2. With the `REGION` attribute you can specify an uncertainty set using one of the predefined uncertainty sets `Box`, `ConvexHull` or `Ellipsoid`. The `UNCERTAINTY` attribute specifies a relationship between the uncertain parameter at hand, and one or more other (uncertain) parameters in your model. The `UNCERTAINTY` and `REGION` attributes are not exclusive, i.e., you are allowed to specify both, in which case AIMMS' generation process of the robust counterpart will make sure that both conditions are satisfied by the final solution.

*The UNCERTAINTY
and REGION
attributes*

The `DISTRIBUTION` attribute is available if the parameter at hand has been declared random using the `Random` property. Random parameters are used by AIMMS' robust optimization framework, and are discussed in full detail in Section 20.3. With the `DISTRIBUTION` attribute you can declare that the values for the random parameter at hand adhere to one of the predefined distributions discussed in Section 20.3.

*The
DISTRIBUTION
attribute*